Binomial Option Pricing, the Black-Scholes Option Pricing Formula, and Exotic Options

The authors show that in the limit the binomial option pricing model considered in their first article converges to the Black-Scholes option pricing formula. They then consider the pricing of "exotic" options, whose value depends on the price path of the underlying asset.

by Simon Benninga and Zvi Wiener

In the first article in this series we introduced the binomial option pricing formula. Since the publication of that article, the Nobel prize in economics has been awarded to Robert Merton and Myron Scholes; along with the late Fischer Black, these two economists helped shape much of the modern pricing theory of options. We start this article by explaining the Black-Scholes option pricing formula; we then use *Mathematica* to show numerically that the binomial price converges to the Black-Scholes price for European options and for some American options. Finally, we consider binomial approaches to pricing "exotic" options–options whose price depends on the whole price path of the underlying asset.

THE BLACK-SCHOLES OPTION PRICING FORMULA

In a path-breaking paper, Black-Scholes (1973) proved the following theorem:

THEOREM. *Consider a European call option on a stock whose current price is S. Suppose that the stock price is lognormally distributed with volatility σ, that the option's exercise price is X, that the exercise date of the option is T, and that the continuously compounded interest rate is r. Furthermore assume that the stock will pay no dividends before the option exercise date T. Then the call price is given by:*

$$C = SN(d_1) - Xe^{-rT}N(d_2),$$

where

$$d_1 = \frac{\ln\left(\frac{S}{X}\right) + \left(r + \frac{\sigma^2}{2}\right)T}{\sigma\sqrt{T}} \quad \text{and} \quad d_2 = d_1 - \sigma\sqrt{T}$$

and where $N()$ indicates values of the cumulative standard normal distribution.

By the *Put-Call Parity Theorem*, suppose we are pricing a European call and put, both with exercise date T and both defined on a stock whose current price is S and which does not pay dividends before the maturity date T. Then if the continuously compounded interest rate is r, the value P of the put, the value C of the call, and the stock price S are related by $P = C + Xe^{-rT} - S$. Applying this theorem, it follows that the Black-Scholes price of a European put written on the same stock with the same exercise price and date is given by $P = -SN(-d_1) + Xe^{-rT}N(-d_2)$.

The only problem with implementing the Black-Scholes formula in *Mathematica* is how to compute values of the standard normal distribution. We offer 4 solutions and suggest trying them all. The four functions `snormal` are different in speed of computation and other properties. Note that the fourth function is based on the `NormalDistribution` function which is defined in the standard package `Statistics`.

```
In[1]:=
snormal1[x_]:=
  Integrate[Exp[-z^2/2]/Sqrt[2*Pi],
   {z,-Infinity,x}]//N
snormal2[x_]:=
  NIntegrate[Exp[-z^2/2]/Sqrt[2*Pi],
   {z,-Infinity,x}];
snormal3[x_]:= Erf[x/Sqrt[2]]/2+0.5;

In[2]:= Needs["Statistics`Master`"]
        ndist=NormalDistribution[0,1];

In[3]:= snormal4[x_]:=CDF[ndist,x]//N;
```

To compare performance, we can calculate 100 values of the normal distribution for each function:

```
In[4]:=
Timing[Table[snormal1[i/100], {i, 100}]; ][[1]]
Timing[Table[snormal2[i/100], {i, 100}]; ][[1]]
Timing[Table[snormal3[i/100], {i, 100}]; ][[1]]
Timing[Table[snormal4[i/100], {i, 100}]; ][[1]]

Out[4]= 95.3 Second
        10.6 Second
        0.05 Second
        0.27 Second
```

The results show how much faster the calculation can be performed with built-in functions (the fourth definition is almost 400 times faster than the first). However, there are some more subtle differences that can be relevant when we try to differentiate these functions. For example, the function `snormal2` is a numerical expression, while the others are analytic.

Having defined the normal distribution, we program the Black-Scholes formula for both puts and calls (note the trick used to switch between the various definitions of the normal distribution):

```
In[5]:= Clear[snormal, d1, d2, bsCall, bsPut]
        snormal=snormal3;
        d1[s_, x_, sigma_, T_, r_]:=
          (Log[s/x]+(r+sigma^2/2)*T)/
           (sigma*Sqrt[T])
        d2[s_, x_, sigma_, T_, r_]:=
          d1[s, x, sigma, T, r]-sigma*Sqrt[T]
        bsCall[s_, x_, sigma_, T_, r_]:=
          s*snormal[d1[s,x,sigma,T,r]]-x*
          Exp[-r*T]*snormal[d2[s,x,sigma,T,r]]
        bsPut[s_,x_,sigma_,T_,r_]:=
          bsCall[s,x,sigma,T,r]+x*Exp[-r*T]-s
```

We can easily use *Mathematica* to graph these functions:

```
In[6]:=
 Plot[bsCall[s, 50, 0.2, 0.5, 0.05],
   {s, 0.01, 70}, PlotRange→All,
   AxesLabel→{"stock price", "call price"}]
```



THE BINOMIAL OPTION PRICE CONVERGES TO THE BLACK-SCHOLES PRICE

In the previous article in this series we defined a binomial option pricing formula for European options for which – in the limit – the stock's price process converges to the lognormal price:

```
In[7]:= Clear[EuropeanOption, EuropeanCall,
         EuropeanPut]
        EuropeanOption[s_, sigma_,
             T_, r_, exercise_Function, n_] :=
          Module[{
           u = N[Exp[Sqrt[T/n]*sigma]],
           d = N[Exp[-Sqrt[T/n]*sigma]],
           R = N[Exp[r*T/n]]},
           p = (R - d)/(R*(u - d));
           q = (u - R)/(R*(u - d));
   Sum[exercise[s*u^j*d^(n - j)]*
    Binomial[n, j]*p^j*q^(n - j),
     {j, 0, n}]]
EuropeanCall[s_, x_, sigma_, T_, r_, n_] :=
   EuropeanCall[s, x, sigma, T, r, n] =
    EuropeanOption[s, sigma, T, r,
     Max[#1 - x, 0] & , n];
  EuropeanPut[s_, x_, sigma_, T_, r_, n_] :=
   EuropeanPut[s, x, sigma, T, r, n] =
    EuropeanOption[s, sigma, T, r,
     Max[x - #1, 0] & , n];
```

We can now show that this binomial pricing formula converges to the Black-Scholes price:

```
In[8]:= ListPlot[
       Table[{n,bsCall[50,45,
        0.4, 0.25, 0.06] -
         EuropeanCall[50, 45, 0.4, 0.25,
          0.06, n]}, {n, 10, 500, 10}],
       PlotJoined→True, PlotRange→All,
       AxesLabel→{n,""}];
```



In the same article, we also defined the following function to price American options:

```
In[9]:=
 AmericanOption[s_,sigma_,T_, r_,
             exercise_Function,n_]:=
  Module[{u=N[ Exp[ Sqrt[T/n]*sigma]],
          d=N[ Exp[-Sqrt[T/n]*sigma]],
  R=N[ Exp[ r* T/n] ], p, q, OpRecurse},
  p=(R-d)/(R*(u-d)); q=(u-R)/(R*(u-d));
  OpRecurse[node_, level_]:=
    OpRecurse[node, level]=
     If[ level==n,
      exercise[s*d^node*u^(level-node)],
  Max[{p,q}.{OpRecurse[node, level+1],
  OpRecurse[node+1, level+1]},
  exercise[s*d^node*u^(level-node)]
 ]];
  OpRecurse[0,0]];
 AmericanCall[ s_, x_, sigma_, T_, r_, n_ ]:=
  AmericanCall[ s, x, sigma, T, r, n ]=
   AmericanOption[ s, sigma, T, r,
     Max[#-x,0]&, n];
 AmericanPut[ s_, x_, sigma_, T_, r_, n_ ]:=
  AmericanPut[ s, x, sigma,T, r, n ]=
  AmericanOption[ s, sigma, T, r,
     Max[x-#,0]&, n];
```

A well-known theorem in the option pricing literature states that the price of an American call on a stock without dividends (as is the case here) is the same as the price of a European option. However, it need not be that an American put and a European put have the same price. Both of these properties are illustrated below:

```
In[10]:=
 Table[{n,AmericanCall[50,45,0.4,0.25,0.06,n]-
  EuropeanCall[50,45,0.4,0.25,0.06,n]},
   {n,1,100,10}]
In[11]:=
 a=Table[{n,AmericanPut[50,45,0.4,1,0.10,n]-
   EuropeanPut[50,45,0.4,1,0.10,n]},
 {n,10,80,10}];
In[12]:=
 ListPlot[a,
  PlotStyle→PointSize[0.02],
  Frame→{True,True,False,False},
  FrameLabel→{"n = iterations",difference},
  PlotLabel→"Difference Between American
               and European Put",
  DefaultFont→{"Helvetica",9} ];
```



It follows that the early exercise option of the American call is worthless (and hence, that the Black-Scholes formula can price both European and American calls on stocks which pay no dividends before the call's maturity); however, the early exercise feature of American puts is potentially valuable.

EXOTIC OPTIONS

European and American puts and calls are by far the most popular types of financial options. However, the development of financial markets has spawned many other types of options. Often these options are marketed as part of a financial package, such as the option implicit in a callable bond or the prepayment option in a mortgage. Most of these options are path dependent and in most cases there is no analytic solution for the price.

In this section we give examples of several kinds of such "exotic" options, together and show how the prices may be calculated numerically with *Mathematica.*

Asian options

An Asian call option gives its owner the right to buy (or sell, if it is an Asian put) a share of stock for the average price during some period between the beginning of the contract and the exercise date of the option. Options on oil, for example, commonly tie the exercise price of the option to the average price of a barrel of oil in the month before the option exercise date. Such an option is useful to a company which buys oil on a monthly basis and wants to protect itself from losing money during periods of high price volatility.

To price an Asian option, we can perform the standard backward induction using a binomial option pricing model. At maturity, the option's payoff is known and is equal to the payoff which results from the average stock price over the whole period. If early exercise of the option is allowed, then one period before exercise, the option's value is equal to the maximum of the price of the option exercised immediately and the present value–using the state prices p and q–of the next-period payoffs (this is the standard American option pricing problem explained in our previous article). The following routine shows how to do this in *Mathematica*:

```
In[13]:=
 Clear[AsianOption,AmerAvgCall,
        EurAvgCall,mean]
 mean[x_List]:= Apply[Plus,x]/Length[x]//N
 AsianOption[s_, sigma_, T_,r_,
  exerciseAtT_Function,
              earlyExercise_Function, n_]:=
  Module[{u=N[Exp[Sqrt[T/n]*sigma]],
   d=N[Exp[-Sqrt[T/n]*sigma]],
   R=N[ Exp[ r*T/n] ], p, q, OpRecurse},
   p=(R-d)/(R*(u-d)); q=(u-R)/(R*(u-d));
  OpRecurse[prices_List, level_]:=
   If[ level==n, exerciseAtT[prices],
  Max[{OpRecurse[Append[ prices,
    u Last[prices]],level+1],
  OpRecurse[Append[ prices,
    d Last[prices]], level+1]}.{p,q},
     earlyExercise[prices]]];
  OpRecurse[{s}, 0]
 ];
In[14]:= AmerAvgCall[s_,sigma_,T_,r_,n_]:=
          AmerAvgCall[s,sigma,T,r,n]=
           AsianOption[s,sigma,T,r,
            Max[0, mean[#]-Last[#]]&,
            Max[0, mean[#]-Last[#]]&,n];
In[15]:= EurAvgCall[s_,sigma_,T_,r_,n_]:=
          EurAvgCall[s,sigma,T,r,n]=
           AsianOption[s,sigma, T, r,
            Max[0, mean[#]-Last[#]]&, 0&,n];
In[16]:= AmerAvgPut[s_,sigma_,T_,r_,n_]:=
          AmerAvgPut[s,sigma,T,r,n]=
           AsianOption[s,sigma, T,r,
            Max[0,Last[#]-mean[#]]&,
            Max[0, Last[#]-mean[#]]&,n];
In[17]:= EurAvgPut[s_,sigma_,T_,r_,n_]:=
          EurAvgPut[s,sigma,T,r,n]=
           AsianOption[s, sigma, T, r,
            Max[0, Last[#]-mean[#]]&, 0&,n];
```

To check the difference between the two call options we use the following example (here the initial stock price is \$100, there are 10 periods, σ= 35% there is one-half year until maturity and the annual risk free interest rate is 8%).

```
In[18]:= s=100;sigma=0.35;T=0.5;r=0.08;n=10;
         AmerAvgCall[s,sigma,T,r,n]
         EurAvgCall[s,sigma,T,r,n]
         AmerAvgPut[s,sigma,T,r,n]
         EurAvgPut[s,sigma,T,r,n]
```

```
Out[18]= 6.38749
         4.6354
         7.82452
         6.60769
```

It is no longer true that an American average call and a European average call have the same price: As the following graphic shows, the American average call is, in general, more valuable than the European average call:

```
In[19]:= Clear[s,sigma,T,r,n];
         s=100;sigma=0.35;r=0.08;n=5;
         Plot[AmerAvgCall[s,sigma,T,r,n]-
          EurAvgCall[s,sigma,T,r,n],{T,0,1},
          Frame→{True,True,False,False},
          PlotLabel→"Difference between\n
                 AmerAvgCall and EurAvgCall",
          FrameLabel→{"T=time to maturity",""},
          DefaultFont→{"Helvetica",10}]
```



REFERENCES

Benninga, Simon. *Financial Modeling*. MIT Press (1997).

COX, J. C., ROSS, S. A., RUBINSTEIN, M. (1979). "Option Pricing: A Simplified Approach," *Journal of Financial Economics*, 7, pp. 229–263.

HULL J., *Options, Futures, and Other Derivatives*, third edition, Prentice Hall, 1997.

OMBERG, E.. "A Note On The Convergence Of Binomial-Pricing And Compound-Option Models," *Journal of Finance*, 1987, v42(2), 463–469.

ABOUT THE AUTHORS

The authors acknowledge grants from Wolfram Research, the Krueger and Eshkol Centers at the Hebrew University, and the Israeli Academy of Science. Wiener's research has benefited from a grant from the Israel Foundations Trustees and the Alon Fellowship.

Simon Benninga is professor of finance at Tel-Aviv University (Israel) and the Wharton School of the University of Pennsylvania. He is the author of *Financial Modeling* (MIT Press, 1997) and of *Corporate Finance: A Valuation Approach* (with Oded Sarig, McGraw-Hill, 1997); he is also the editor of the *European Finance Review*.

Simon Benninga
Faculty of Management
Tel-Aviv University, Tel-Aviv, Israel
benninga@post.tau.ac.il
http://finance.wharton.upenn.edu/~benninga

Zvi Wiener is an assistant professor of Finance at Business School of the Hebrew University of Jerusalem. He is currently visiting professor of finance at the Olin School of Business at Washington University in St. Louis. His finance research concentrates on the pricing of derivative securities, value-at-risk, computational finace and stochastic dominance.

Zvi Wiener
Finance Department, Business School
Hebrew University, Jerusalem, Israel
mswiener@pluto.mscc.huji.ac.il
http://pluto.mscc.huji.ac.il/~mswiener/zvi.html

ELECTRONIC SUBSCRIPTIONS

Included in the distribution for each electronic subscription is the file `Black-Scholes.nb` containing *Mathematica* code for the material described in this article.